Web Appendix: Sample input data, SAS code, and output for the maximized sequential probability ratio test using a binomial probability model to implement self-controlled case series and difference-in-difference analyses.

Web Table 1: Input data for sample outcome:

Week start date	Cumulative number of 1st TIV doses	`cumDxVac` (cumulative number of events in risk window)	`cumPC` (cumulative number of events in control window)
7-Sep-08	898	0	1
28-Sep-08	48,023	1	1
12-Oct-08	190,043	2	8
19-Oct-08	377,943	2	14
26-Oct-08	523,936	2	17
2-Nov-08	1,067,548	14	47
9-Nov-08	1,161,599	15	52
16-Nov-08	1,292,667	17	54
23-Nov-08	1,363,612	19	57
30-Nov-08	1,421,750	19	61
7-Dec-08	1,477,088	20	67
14-Dec-08	1,560,261	21	68

Variable definitions for SAS code:

`data1` = Dataset name

`ratio` = Ratio of number of days in risk window vs. control window for sample outcome: 0.35

`&b/&a` = Number of events in risk window vs. control window of cumulative previous seasons for sample outcome = 0.258

SAS code:

```
*******************************************
Self-controlled case series
*******************************************;
data SCCS;
     set data1;
     format relativeRisk logLikelihoodRatio 6.4;
     cumtotal = cumPC + cumDxVac;
     if cumPC gt 0 then relativeRisk = cumDxVac /(cumPC *
     ratio);
     if cumDxVac gt 0 and cumTotal gt 0 then lla = cumDxVac *
        log(cumDxVac/cumTotal);
     else lla = 0;
     if cumTotal gt cumDxVac then llb = cumPC * log(1 -
        cumDxVac/cumTotal);
     else llb = 0;
     llc = - cumDxVac * log(1 / (1/ratio + 1));
     lld = - cumPC * log(1- (1 / (1/ratio + 1)));
     if (cumDxVac * 1/ratio) gt cumPC then logLikelihoodRatio =
        sum(lla,llb,llc,lld);
     else logLikelihoodRatio = 0;
run;
```

```
*****************************************
Difference-in-difference
*****************************************;
data DiD;
     set data1;
     format relativeRisk logLikelihoodRatio 6.4;
     cumtotal = cumPC + cumDxVac;
     %let d = cumDxVac;
     %let c = cumPC;
     if cumPC gt 0 and &a gt 0 then relativeRisk = (cumDxVac
       /cumPC)/(&b/&a);
     %let currentSum = sum(&c,&d);
     %let pastSum = sum(&a,&b);
     currentSum = sum(&c,&d);
     pastSum = sum(&a,&b);
     if &currentSum gt 0 then do;
          lla = &d * log(&d/&currentSum);
          llb = &c * log(&c/&currentSum);
     end;
     else do;
          lla = 0;
          llb = 0;
     end;
     if &pastSum gt 0 then do;
          llc = - &d * log(&b/&pastSum);
          lld = - &c * log(&a/&pastSum);
     end;
     else do;
          llc = 0;
          lld = 0;
     end;
     if relativeRisk gt 1 then logLikelihoodRatio =
     sum(lla,llb,llc,lld);
     else logLikelihoodRatio = 0;
run;
```

Web Table 2: SCCS sample output:

Week start date	Relative risk	Log-likelihood ratio
7-Sep-08	0	0
28-Sep-08	2.8571	0.2637
12-Oct-08	0.7143	0
19-Oct-08	0.4082	0
26-Oct-08	0.3361	0
2-Nov-08	0.8511	0
16-Nov-08	0.8242	0
9-Nov-08	0.8995	0
23-Nov-08	0.9524	0
30-Nov-08	0.8899	0
30-Nov-08	0.8529	0
14-Dec-08	0.8824	0

Web Table 3: DiD sample output:

Week start date	Relative risk ratio	Log-likelihood ratio
7-Sep-08	0	0
28-Sep-08	3.8686	0.4264
12-Oct-08	0.9672	0
19-Oct-08	0.5527	0
26-Oct-08	0.4551	0
2-Nov-08	1.1524	0.1057
16-Nov-08	1.1159	0.0686
9-Nov-08	1.2179	0.2426
23-Nov-08	1.2895	0.4409
30-Nov-08	1.205	0.2436
30-Nov-08	1.1548	0.1554
14-Dec-08	1.1947	0.2459